
3GPP TSG CT4 Meeting #83
C4-182387
3GPP TSG CT3 Meeting #95
C3-181351
Montreal, CANADA; 26th Feb – 02nd Mar 2018

Source:
Nokia, Nokia Shanghai-Bell, Oracle, AT&T
Title:
Word Macro to Convert Data Model to OpenApi
Agenda item:
6.2.1.4 (CT4) / 15.2 (CT3)
Document for:
Decision

Discussion
This document contains a Word Macro (both embedded and shown in the Annex) to convert Data Model Clauses of Specs following the agreed template to OpenAPI schema definitions.

The Output is suitable for the components/schemas part of an OpenAPI file.

The proposals in C4-182111 and C4-182226 are also assumed.

The Macro has already been used to genaerate the OpenAPI files proposed by Nokia in this meeting.

Interested companies are invited to test and use the Macro as a help for providing OpenAPI Annexes and to report possible bugs or improvement proposals.

The Macro is also suitable for T8 TS 29.122 in CT3.
Usage
Configure Word to Show the "Developer" Tab.

Copy the Text in the Annex of this contribution

Open the Visual Basic Editor using the Developer Tab

Add the text to the "Normal" Macros.

1If NreusedTypes > 0 Then

NreusedTypes = NreusedTypes - 1
text = reusedTypes (0)
For IreusedType = 1 To NreusedTypes

text = text + " " + reusedTypes (IreusedType)
Next
Ivar = 0
For Each aVar In ActiveDocument.Variables

If avVar.name = "CommonDataTypes" Then @

Ivar = aVar.Index

Alternatively, you can open the file with data to import and add the Macros that file.

Execute the Macros using the "Macros" button of the developer Tab

The following Macros are defined:

Convert_Types_to_OpenAPI:
Converts the selected text into an OpenAPI fragment. Text with Data Models is suitable to be marked, e.g. the Clause 5.2.1 of TS 29.122 (for Common Data Types) or any Clause 5.X.1 (for data Types of a given API). For data types of an API, the Mark_Common_Data_Types Macro should be executed before.
A corresponding OpenAPI fragment will be inserted directly below the selected text. The OpenAPI fragment is suitable for the components/schemas part of an OpenAPI file.

The following Macros are only applicable for T8:

Mark_Common_Data_Types

Reads the data types within the selected text into an OpenAPI fragment. Those data types are afterwards regarded as Common Data Types and will be referenced accordingly when the Convert_Types_to_OpenAPI is afterwards executed. Text with Common Data Models is suitable to be marked, i.e. the Clause 5.2.1 of TS 29.122.
Clear_Common_Data_Types

Cleares any Common Data types previously marked with the Mark_Common_Data_Types Macro. This is required before the Convert_Types_to_OpenAPI is used to convert the Common Data types in Clause 5.2.1 of TS 29.122.

Annex:
Sub Convert_Types_to_OpenAPI()

'

' 02/3/2018 Th Belling

'

    Dim arow As Row

    Dim atable, tables() As Table

    Dim text, destext, tablecaptions(), reusedTypes(), refForReusedTypes(), _

        name, dataType, requiredAtts(), LowerBoundary, UpperBoundary, ref, _

        TSname As String

    Dim Iarow, Iparagraph, Nparagraph, Itable, Ntable, Itable2, Iformat, _

        Irequired, Iboundary, NreusedTypes, IreusedTypes, Ivar, I As Integer

    Dim isBasicType, isMap, isArray, LowerBoundaryExists, UpperBoundaryExists, _

        reusedTypesExist, PreColumnExists, SeparateTypeExists As Boolean

    Dim aVar As Variable

    Ntable = Selection.Range.tables.Count

    If Ntable = 0 Then

        MsgBox "Please select at least one Table !"

        Exit Sub

    End If

    NreusedTypes = 0

    Ivar = 0

    For Each aVar In ActiveDocument.Variables

        If aVar.name = "CommonDataTypes" Then

            Ivar = aVar.Index

            Exit For

        End If

    Next aVar

    If Ivar > 0 Then

        I = 0

        For Each aVar In ActiveDocument.Variables

            If aVar.name = "TSName" Then

                I = aVar.Index

                Exit For

            End If

        Next aVar

        If I > 0 Then

            TSname = ActiveDocument.Variables(I).Value

        Else

            TSname = ""

        End If

        TSname = InputBox("TS Number, e.g. TS29122 ?", "References to Common Data", TSname)

        If Len(TSname) = 0 Then

            Exit Sub

        End If

        If I = 0 Then

            ActiveDocument.Variables.Add name:="TSName", Value:=TSname

        Else

            ActiveDocument.Variables(I).Value = TSname

        End If

        text = ActiveDocument.Variables(Ivar).Value

        I = InStr(text, " ")

        While I > 0

            name = Left(text, I - 1)

            ReDim Preserve reusedTypes(NreusedTypes)

            reusedTypes(NreusedTypes) = name

            ReDim Preserve refForReusedTypes(NreusedTypes)

            refForReusedTypes(NreusedTypes) = _

                "'" + TSname + "_CommonData.yaml#/components/schemas/" + name + "'"

            text = Right(text, Len(text) - I)

            NreusedTypes = NreusedTypes + 1

            I = InStr(text, " ")

        Wend

        ReDim Preserve reusedTypes(NreusedTypes)

        reusedTypes(NreusedTypes) = text

        NreusedTypes = NreusedTypes + 1

    End If

    Itable = 0

    Itable2 = 0

    Nparagraph = Selection.Paragraphs.Count

    For Iparagraph = 1 To Nparagraph

        text = Selection.Paragraphs(Iparagraph).Range.text

        If InStr(text, "Table") = 1 And Selection.Paragraphs(Iparagraph).Range.ParagraphStyle = "TH" Then

            Itable = Itable + 1

            If Itable > Ntable Then

                MsgBox "Error: to many table captions found"

                Exit Sub

            End If

            If InStr(text, "Enumeration ") > 1 Or _

                InStr(text, "enumeration ") > 1 Or _

                InStr(text, "Definition of type ") > 1 Or _

                InStr(text, "definition of type ") > 1 Or _

                InStr(text, "Simple Data Types") > 1 Or _

                InStr(text, "Simple data types") > 1 Or _

                InStr(text, "simple data types") > 1 Or _

                ((InStr(text, "Definition of ") > 1 Or _

                InStr(text, "definition of ") > 1) And _

                (InStr(text, "data Type") > 1 Or _

                InStr(text, "Data Type") > 1 Or _

                InStr(text, "data type") > 1)) Then

                ReDim Preserve tables(Itable2)

                Set tables(Itable2) = Selection.Range.tables(Itable)

                ReDim Preserve tablecaptions(Itable2)

                text = Left(text, Len(text) - 1)

                tablecaptions(Itable2) = text

                Itable2 = Itable2 + 1

            ElseIf InStr(text, "re-used Data Types") > 1 Then

                Iarow = 1

                For Each arow In Selection.Range.tables(Itable).Rows

                    If Iarow > 1 Then

                        If arow.Cells.Count > 1 And Len(arow.Cells(1).Range.text) > 2 Then

                            ReDim Preserve reusedTypes(NreusedTypes)

                            ReDim Preserve refForReusedTypes(NreusedTypes)

                            text = arow.Cells(1).Range.text

                            name = Left(text, Len(text) - 2)

                            reusedTypes(NreusedTypes) = name

                            text = arow.Cells(2).Range.text

                            text = Replace(text, "3GPP", "")

                            text = Replace(text, " ", "")

                            text = Replace(text, " ", "")

                            text = Replace(text, ".", "")

                            text = "'" + Left(text, 7) + _

                                "_CommonData.yaml#/components/schemas/" + _

                                name + "'"

                            refForReusedTypes(NreusedTypes) = text

                            NreusedTypes = NreusedTypes + 1

                        End If

                    End If

                    Iarow = Iarow + 1

                Next arow

            End If

        End If

    Next

    Ntable = Itable2 - 1

    If Ntable = -1 Then

        MsgBox "Error: No suitable Table found. Did you mark the table heading?"

        Exit Sub

    End If

    If NreusedTypes > 0 Then

        NreusedTypes = NreusedTypes - 1

        reusedTypesExist = True

    Else

        reusedTypesExist = False

    End If

    Selection.Collapse Direction:=wdCollapseEnd

    Selection.InsertParagraphBefore

    Selection.Collapse Direction:=wdCollapseStart

    Selection.Range.Paragraphs(1).Style = "PL"

    For Itable = 0 To Ntable

        Set atable = tables(Itable)

        text = tablecaptions(Itable)

        If InStr(text, "Enumeration ") > 1 Or InStr(text, "enumeration ") > 1 Then

            If InStr(text, "Enumeration ") > 1 Then

                text = Right(text, Len(text) - InStr(text, "Enumeration ") - 11)

            Else

                text = Right(text, Len(text) - InStr(text, "enumeration ") - 11)

            End If

            text = "    " + text + ":"

            Selection.TypeText (text)

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Selection.TypeText ("      type: string")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Selection.TypeText ("      enum:")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Iarow = 1

            For Each arow In atable.Rows

                If Iarow > 1 Then

                    If arow.Cells.Count > 1 And Len(arow.Cells(1).Range.text) > 2 Then

                        text = arow.Cells(1).Range.text

                        text = Left(text, Len(text) - 2)

                        text = "        - " + text

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                    End If

                End If

                Iarow = Iarow + 1

            Next arow

            Selection.TypeText ("      description: >")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Iarow = 1

            For Each arow In atable.Rows

                If Iarow > 1 Then

                    If arow.Cells.Count > 1 And Len(arow.Cells(1).Range.text) > 2 Then

                        text = arow.Cells(2).Range.text

                        text = Replace(text, Chr(13), " ")

                        text = Replace(text, Chr(11), " ")

                        text = Replace(text, ":", " ")

                        destext = Left(text, Len(text) - 2)

                        text = arow.Cells(1).Range.text

                        text = Left(text, Len(text) - 2)

                        text = "        - " + text + ": " + destext

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                    End If

                End If

                Iarow = Iarow + 1

            Next arow

        ElseIf InStr(text, "Simple Data Types") > 1 _

            Or InStr(text, "Simple data types") > 1 _

            Or InStr(text, "simple data types") > 1 Then

            SeparateTypeExists = (InStr(atable.Rows(1).Cells(2).Range.text, "Type Definition") = 1)

            Iarow = 1

            For Each arow In atable.Rows

                If Iarow > 1 Then

                    If arow.Cells.Count > 1 And Len(arow.Cells(1).Range.text) > 2 Then

                        text = arow.Cells(1).Range.text

                        text = Left(text, Len(text) - 2)

                        text = "    " + text + ":"

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                        text = arow.Cells(2).Range.text

                        text = Left(text, Len(text) - 2)

                        If SeparateTypeExists Then

                            If InStr(text, "string") = 1 Or InStr(text, "integer") = 1 _

                                Or InStr(text, "number") = 1 Or InStr(text, "boolean") = 1 _

                                Or InStr(text, "FFS") = 1 Then

                                text = "      type: " + text

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                                text = arow.Cells(3).Range.text

                                Iformat = InStr(text, "format shall be designated as ")

                                If Iformat > 1 Then

                                    text = Right(text, Len(text) - Iformat - 30)

                                    Iformat = InStr(text, Chr(34))

                                    If Iformat > 1 Then

                                        text = Left(text, Iformat - 1)

                                        text = "      format: " + text

                                        Selection.TypeText (text)

                                        Selection.InsertParagraphAfter

                                        Selection.Collapse Direction:=wdCollapseEnd

                                    End If

                                Else

                                    Iformat = InStr(text, "with format ")

                                    If Iformat > 1 Then

                                        text = Right(text, Len(text) - Iformat - 12)

                                        Iformat = InStr(text, Chr(34))

                                        If Iformat > 1 Then

                                            text = Left(text, Iformat - 1)

                                            text = "      format: " + text

                                            Selection.TypeText (text)

                                            Selection.InsertParagraphAfter

                                            Selection.Collapse Direction:=wdCollapseEnd

                                        End If

                                    End If

                                End If

                            Else

                                ref = "'#/components/schemas/" + text + "'"

                                If reusedTypesExist Then

                                    For IreusedTypes = 0 To NreusedTypes

                                        If reusedTypes(IreusedTypes) = text Then

                                            ref = refForReusedTypes(IreusedTypes)

                                            Exit For

                                        End If

                                    Next

                                End If

                                text = "      $ref: " + ref

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                            text = arow.Cells(3).Range.text

                            text = Left(text, Len(text) - 2)

                        Else

                            If InStr(text, "string") = 1 Or InStr(text, "String") = 1 Then

                                dataType = "string"

                            ElseIf InStr(text, "integer") = 1 Or _

                                InStr(text, "Integer") = 1 Or _

                                InStr(text, "Unsigned integer") = 1 Or _

                                InStr(text, "unsigned integer") = 1 Then

                                dataType = "integer"

                            ElseIf InStr(text, "number") = 1 Or _

                                InStr(text, "Number") = 1 Then

                                dataType = "number"

                            ElseIf InStr(text, "boolean") = 1 Or _

                                InStr(text, "Boolean") = 1 Then

                                dataType = "boolean"

                            Else

                                dataType = "???"

                            End If

                            Selection.TypeText ("      type: " + dataType)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                        End If

                        text = Replace(text, Chr(13), " ")

                        text = Replace(text, Chr(11), " ")

                        text = Replace(text, ":", " ")

                        text = "      description: " + text

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                    End If

                End If

                Iarow = Iarow + 1

            Next arow

         ElseIf InStr(text, "as a list of alternatives") > 1 Then

            If InStr(text, "Definition of type ") > 1 Then

                text = Right(text, Len(text) - InStr(text, "Definition of type ") - 18)

            Else

                text = Right(text, Len(text) - InStr(text, "definition of type ") - 18)

            End If

            text = Left(text, Len(text) - 26)

            text = "    " + text + ":"

            Selection.TypeText (text)

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Selection.TypeText ("      oneOf:")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Iarow = 1

            For Each arow In atable.Rows

                If Iarow > 1 Then

                    If arow.Cells.Count > 2 And Len(arow.Cells(1).Range.text) > 2 Then

                        isBasicType = False

                        isMap = False

                        isArray = False

                        LowerBoundaryExists = False

                        UpperBoundaryExists = False

                        text = arow.Cells(1).Range.text

                        dataType = Left(text, Len(text) - 2)

                        If InStr(dataType, "array(") = 1 Then

                            dataType = Left(dataType, Len(dataType) - 1)

                            dataType = Right(dataType, Len(dataType) - 6)

                            isArray = True

                        ElseIf InStr(dataType, "map(") = 1 Then

                            dataType = Left(dataType, Len(dataType) - 1)

                            dataType = Right(dataType, Len(dataType) - 4)

                            isMap = True

                        End If

                        If InStr(dataType, "string") = 1 Or InStr(dataType, "integer") = 1 _

                            Or InStr(dataType, "number") = 1 Or InStr(dataType, "boolean") = 1 Then

                            isBasicType = True

                        End If

                        ref = "'#/components/schemas/" + dataType + "'"

                        If reusedTypesExist Then

                            For IreusedTypes = 0 To NreusedTypes

                                If reusedTypes(IreusedTypes) = dataType Then

                                    ref = refForReusedTypes(IreusedTypes)

                                    Exit For

                                End If

                            Next

                        End If

                        text = arow.Cells(2).Range.text

                        text = Left(text, Len(text) - 2)

                        Iboundary = InStr(text, "..")

                        If Iboundary > 1 Then

                            LowerBoundary = Left(text, Iboundary - 1)

                            UpperBoundary = Right(text, Len(text) - Iboundary - 1)

                            LowerBoundaryExists = IsNumeric(LowerBoundary)

                            UpperBoundaryExists = IsNumeric(UpperBoundary)

                        End If

                        If isArray Then

                            Selection.TypeText ("      - type: array")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            Selection.TypeText ("        items:")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If isBasicType Then

                                text = "          type: " + dataType

                            Else

                                text = "          $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If LowerBoundaryExists Then

                                text = "        minItems: " + LowerBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                            If UpperBoundaryExists Then

                                text = "        maxItems: " + UpperBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                        ElseIf isMap Then

                            Selection.TypeText ("      - type: object")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            Selection.TypeText ("        items:")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If isBasicType Then

                                text = "          type: " + dataType

                            Else

                                text = "          $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If LowerBoundaryExists Then

                                text = "        minProperties: " + LowerBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                            If UpperBoundaryExists Then

                                text = "        maxProperties: " + UpperBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                        Else

                            If isBasicType Then

                                text = "      - type: " + dataType

                            Else

                                text = "      - $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                        End If

                        text = arow.Cells(3).Range.text

                        text = Replace(text, Chr(13), " ")

                        text = Replace(text, Chr(11), " ")

                        text = Replace(text, ":", " ")

                        text = Left(text, Len(text) - 2)

                        text = "        description: " + text

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                    End If

                End If

                Iarow = Iarow + 1

            Next arow

        ElseIf InStr(text, "Definition of ") > 1 Or _

            InStr(text, "definition of ") > 1 Then

            PreColumnExists = (InStr(atable.Rows(1).Cells(3).Range.text, "P") = 1)

            If InStr(text, "Definition of type ") > 1 Then

                text = Right(text, Len(text) - InStr(text, "Definition of type ") - 18)

            ElseIf InStr(text, "definition of type ") > 1 Then

                text = Right(text, Len(text) - InStr(text, "definition of type ") - 18)

            Else

                If InStr(text, "Definition of the ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of the ") - 17)

                ElseIf InStr(text, "definition of the ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "definition of the ") - 17)

                ElseIf InStr(text, "Definition of ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of ") - 13)

                Else

                    text = Right(text, Len(text) - InStr(text, "definition of ") - 13)

                End If

                text = Left(text, Len(text) - 11)

            End If

            text = "    " + text + ":"

            Selection.TypeText (text)

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Selection.TypeText ("      type: object")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Selection.TypeText ("      properties:")

            Selection.InsertParagraphAfter

            Selection.Collapse Direction:=wdCollapseEnd

            Iarow = 1

            Irequired = 0

            For Each arow In atable.Rows

                If Iarow > 1 Then

                    If arow.Cells.Count > 3 And Len(arow.Cells(1).Range.text) > 2 Then

                        isBasicType = False

                        isMap = False

                        isArray = False

                        LowerBoundaryExists = False

                        UpperBoundaryExists = False

                        text = arow.Cells(1).Range.text

                        name = Left(text, Len(text) - 2)

                        text = "        " + name + ":"

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                        text = arow.Cells(2).Range.text

                        dataType = Left(text, Len(text) - 2)

                        If InStr(dataType, "array(") = 1 Then

                            dataType = Left(dataType, Len(dataType) - 1)

                            dataType = Right(dataType, Len(dataType) - 6)

                            isArray = True

                        ElseIf InStr(dataType, "map(") = 1 Then

                            dataType = Left(dataType, Len(dataType) - 1)

                            dataType = Right(dataType, Len(dataType) - 4)

                            isMap = True

                        End If

                        If InStr(dataType, "string") = 1 Or InStr(dataType, "integer") = 1 _

                            Or InStr(dataType, "number") = 1 Or InStr(dataType, "boolean") = 1 Then

                            isBasicType = True

                        End If

                        ref = "'#/components/schemas/" + dataType + "'"

                        If reusedTypesExist Then

                            For IreusedTypes = 0 To NreusedTypes

                                If reusedTypes(IreusedTypes) = dataType Then

                                    ref = refForReusedTypes(IreusedTypes)

                                    Exit For

                                End If

                            Next

                        End If

                        text = arow.Cells(3).Range.text

                        text = Left(text, Len(text) - 2)

                        If PreColumnExists Then

                            If InStr(text, "M") = 1 Then

                                ReDim Preserve requiredAtts(Irequired)

                                requiredAtts(Irequired) = name

                                Irequired = Irequired + 1

                            End If

                            text = arow.Cells(4).Range.text

                            text = Left(text, Len(text) - 2)

                        End If

                        Iboundary = InStr(text, "..")

                        If Iboundary > 1 Then

                            LowerBoundary = Left(text, Iboundary - 1)

                            UpperBoundary = Right(text, Len(text) - Iboundary - 1)

                            LowerBoundaryExists = IsNumeric(LowerBoundary)

                            UpperBoundaryExists = IsNumeric(UpperBoundary)

                            If Not PreColumnExists And (InStr(LowerBoundary, "1") = 1) Then

                                ReDim Preserve requiredAtts(Irequired)

                                requiredAtts(Irequired) = name

                                Irequired = Irequired + 1

                            End If

                        Else

                            If Not PreColumnExists And (InStr(text, "1") = 1) Then

                                ReDim Preserve requiredAtts(Irequired)

                                requiredAtts(Irequired) = name

                                Irequired = Irequired + 1

                            End If

                        End If

                        If isArray Then

                            Selection.TypeText ("          type: array")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            Selection.TypeText ("          items:")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If isBasicType Then

                                text = "            type: " + dataType

                            Else

                                text = "            $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If LowerBoundaryExists Then

                                text = "          minItems: " + LowerBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                            If UpperBoundaryExists Then

                                text = "          maxItems: " + UpperBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                        ElseIf isMap Then

                            Selection.TypeText ("          type: object")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            Selection.TypeText ("          items:")

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If isBasicType Then

                                text = "            type: " + dataType

                            Else

                                text = "            $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                            If LowerBoundaryExists Then

                                text = "          minProperties: " + LowerBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                            If UpperBoundaryExists Then

                                text = "          maxProperties: " + UpperBoundary

                                Selection.TypeText (text)

                                Selection.InsertParagraphAfter

                                Selection.Collapse Direction:=wdCollapseEnd

                            End If

                        Else

                            If isBasicType Then

                                text = "          type: " + dataType

                            Else

                                text = "          $ref: " + ref

                            End If

                            Selection.TypeText (text)

                            Selection.InsertParagraphAfter

                            Selection.Collapse Direction:=wdCollapseEnd

                        End If

                        If PreColumnExists Then

                            text = arow.Cells(5).Range.text

                        Else

                            text = arow.Cells(4).Range.text

                        End If

                        text = Replace(text, Chr(13), " ")

                        text = Replace(text, Chr(11), " ")

                        text = Replace(text, ":", " ")

                        text = Left(text, Len(text) - 2)

                        text = "          description: " + text

                        Selection.TypeText (text)

                        Selection.InsertParagraphAfter

                        Selection.Collapse Direction:=wdCollapseEnd

                    End If

                End If

                Iarow = Iarow + 1

            Next arow

            If Irequired > 0 Then

                Selection.TypeText ("      required:")

                Selection.InsertParagraphAfter

                Selection.Collapse Direction:=wdCollapseEnd

                For Each text In requiredAtts

                    Selection.TypeText ("        - " + text)

                    Selection.InsertParagraphAfter

                    Selection.Collapse Direction:=wdCollapseEnd

                Next text

            End If

        End If

    Next

End Sub

Sub Mark_Common_Data_Types()

'

' 02/3/2018 Th Belling

'

    Dim arow As Row

    Dim atable, tables() As Table

    Dim text, reusedTypes() As String

    Dim Iarow, Iparagraph, Nparagraph, Itable, Ntable, Ivar, NreusedTypes, IreusedType As Integer

    Dim aVar As Variable

    Ntable = Selection.Range.tables.Count

    If Ntable = 0 Then

        MsgBox "Please select at least one Table !"

        Exit Sub

    End If

    Itable = 0

    NreusedTypes = 0

    Nparagraph = Selection.Paragraphs.Count

    For Iparagraph = 1 To Nparagraph

        text = Selection.Paragraphs(Iparagraph).Range.text

        If InStr(text, "Table") = 1 And Selection.Paragraphs(Iparagraph).Range.ParagraphStyle = "TH" Then

            Itable = Itable + 1

            If Itable > Ntable Then

                MsgBox "Error: to many table captions found"

                Exit Sub

            End If

            If InStr(text, "Enumeration ") > 1 Or _

                InStr(text, "enumeration ") > 1 Then

                If InStr(text, "Enumeration ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Enumeration ") - 11)

                Else

                    text = Right(text, Len(text) - InStr(text, "enumeration ") - 11)

                End If

                ReDim Preserve reusedTypes(NreusedTypes)

                reusedTypes(NreusedTypes) = text

                NreusedTypes = NreusedTypes + 1

            ElseIf InStr(text, "as a list of alternatives") > 1 Then

                If InStr(text, "Definition of type ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of type ") - 18)

                Else

                    text = Right(text, Len(text) - InStr(text, "definition of type ") - 18)

                End If

                text = Left(text, Len(text) - 26)

                ReDim Preserve reusedTypes(NreusedTypes)

                reusedTypes(NreusedTypes) = text

                NreusedTypes = NreusedTypes + 1

            ElseIf InStr(text, "Definition of type ") > 1 Or _

                InStr(text, "definition of type ") > 1 Then

                If InStr(text, "Definition of type ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of type ") - 18)

                Else

                    text = Right(text, Len(text) - InStr(text, "definition of type ") - 18)

                End If

                ReDim Preserve reusedTypes(NreusedTypes)

                reusedTypes(NreusedTypes) = text

                NreusedTypes = NreusedTypes + 1

            ElseIf ((InStr(text, "Definition of ") > 1 Or _

                InStr(text, "definition of ") > 1) And _

                (InStr(text, "data Type") > 1 Or _

                InStr(text, "Data Type") > 1 Or _

                InStr(text, "data type") > 1)) Then

                If InStr(text, "Definition of the ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of the ") - 17)

                ElseIf InStr(text, "definition of the ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "definition of the ") - 17)

                ElseIf InStr(text, "Definition of ") > 1 Then

                    text = Right(text, Len(text) - InStr(text, "Definition of ") - 13)

                Else

                    text = Right(text, Len(text) - InStr(text, "definition of ") - 13)

                End If

                text = Left(text, Len(text) - 11)

                ReDim Preserve reusedTypes(NreusedTypes)

                reusedTypes(NreusedTypes) = text

                NreusedTypes = NreusedTypes + 1

            ElseIf InStr(text, "Simple Data Types") > 1 Or _

                InStr(text, "Simple data types") > 1 Or _

                InStr(text, "simple data types") > 1 Then

                Iarow = 1

                For Each arow In Selection.Range.tables(Itable).Rows

                    If Iarow > 1 Then

                        If arow.Cells.Count > 1 And Len(arow.Cells(1).Range.text) > 2 Then

                            ReDim Preserve reusedTypes(NreusedTypes)

                            text = arow.Cells(1).Range.text

                            text = Left(text, Len(text) - 2)

                            reusedTypes(NreusedTypes) = text

                            NreusedTypes = NreusedTypes + 1

                        End If

                    End If

                    Iarow = Iarow + 1

                Next arow

            End If

        End If

    Next

    If Ntable = -1 Then

        MsgBox "Error: No suitable Table found. Did you mark the table heading?"

        Exit Sub

    End If

    If NreusedTypes > 0 Then

        NreusedTypes = NreusedTypes - 1

        text = reusedTypes(0)

        For IreusedType = 1 To NreusedTypes

            text = text + " " + reusedTypes(IreusedType)

        Next

        Ivar = 0

        For Each aVar In ActiveDocument.Variables

            If aVar.name = "CommonDataTypes" Then

                Ivar = aVar.Index

                Exit For

            End If

        Next aVar

        If Ivar = 0 Then

            ActiveDocument.Variables.Add name:="CommonDataTypes", Value:=text

        Else

            ActiveDocument.Variables(Ivar).Value = text

        End If

    End If

End Sub

Sub Clear_Common_Data_Types()

'

' 02/3/2018 Th Belling

'

    Dim Ivar As Integer

    Dim aVar As Variable

    Ivar = 0

    For Each aVar In ActiveDocument.Variables

        If aVar.name = "CommonDataTypes" Then

            Ivar = aVar.Index

            Exit For

        End If

    Next aVar

    If Ivar > 0 Then

        ActiveDocument.Variables(Ivar).Delete

    End If

End Sub

